3. Gyakorlat

Geometriai valószínség

Elfordul, hogy egy esemény kimenetele nem diszkrét értékekbl áll, hanem a hozzá kapcsolt valószínségi változó folytonosan vehet fel értékeket. Ilyenek lehetnek pl. események bekövetkezésének tér- és idkoordinátái, vagy ezzel összefügg adatok. Ezeknek gyakran vagy közvetlen geometriai tartalmuk van, vagy legalábbis közvetetten geometriai modellel tárgyalhatók, melyben a valószínség mint mérték valami geometriai jelleg mértékhez (hosszúság, terület, térfogat) kapcsolható. Ilyenkor beszélünk geometriai valószínségrl ill. valószínségi mezrl.

A valószínség mint geometriai mérték

Válasszunk véletlenszeren egy pontot a [0, 20] intervallumból. Mekkora valószínséggel esik a választott pont az [5, 10] intervallumba?

Mind az összes, mind a kedvez kimenetek száma végtelen, tehát a kombinatorikus valószínségi meznél megszokott lehetség-összeszámlálás majd eset-számok arányítása itt nem mködik.
A modell szerinti kísérlet eredménye ilyesmi:

```
> restart:
  with(Statistics):
  with(plots):
```

```
> X:=RandomVariable(Uniform(0,20)): # véletlen szám a [0;20]
  intervallumból
  minta:=Sample(X, 100):
  pontok:=[seq([minta[i], 0], i=1..100)]:
  display(plot(pontok, style=point), plot(0, x=5..10, color=
  green, thickness=3));
```

Szemlélet alapján érezzük, hogy a véletlenszeren (egyenletes eloszlással) kisorsolt pontok az értékkészletet jelent intervallumon egyenletesen szóródnak el: a kísérletek sokasága során nem látunk tendencia-jelleg srsödéseket, két különálló, egyforma hosszúságú részintervallum esetén nem gondoljuk, hogy bármelyik mellett szólna olyan tényez, hogy rajta tendencia jelleggel több érték jelenne meg (végtelenbe tartó kísérletszám melletti határesetben nagyobb volna a rá esés valószínsége). A valószínségi mérték elkülönült eseményhalmazokra való összegzdésével együtt mindez (szemléletünknek megfelelen, és egzaktul beláthatóan is) azt jelenti, hogy a modellbeli intervallum-szer eseményhalmazok valószínségi mértéke az intervallum-hosszakkal mint geometriai mértékkel azonosíthatók - a teljes eseménytérhez mint alapintervallumhoz mért arány szerint.

Vagyis a geometriai modellben (a klasszikus valószínségi meznél megszokott eshetség-számok arányítása helyett) geometriai mértékeket arányítunk: további paraméterek híján nincs más következetes lehetség, mint úgy tekinteni, hogy a keresett valószínség az [5, 10] és [0, 20] intervallumok mértékének (hosszának) aránya:

```
> p := (10 - 5) / (20 - 0);
```

Érdekes belegondolni, hogy mi a valószínsége egy adott szám, pl. a 12 kisorsolásának. A számegyenesen egy adott számnak mint pontszer objektumnak a geometriai mértéke 0, és a valószínséget is annak kell tekintenünk: nem tudunk olyan kicsi pozitív értéket mondani, amely arányában elvárhatnánk, hogy sok kísérletbl stabilan ebben az arányban ki fog sorsolódni egy konkrét szám. Ugyanakkor elfordulhat, hogy épp kijön, hiszen valami mindig kijön, miközben külön-külön bármely pont kisorsolásának valószínségére ugyanez a zérus igaz. Tehát itt arra látunk példát, hogy egy nulla valószínség esemény (megfelel valószínségi mezben) nem okvetlenül lehetetlen (st, itt az összes elemi esemény nulla valószínség).

A valóságban persze ezek a példák is mindig diszkrét értékkészletek, amennyiben rendszerint egy valódi kísérleti konstrukció (pl. fizikai kísérlet valamely paramétere) értékeit is véges pontossággal mérjük, és a véletlenszám-generátorral intervallumról való sorosolás eredményét is véges pontossággal fejezzük ki. Amilyen egyszer megfogalmazni, hogy "válasszunk egy véletlen értéket a $[0, 1]$ intervallumról", a gyakorlatban ez nem végrehajtható feladat. Az egzakt geometriai modell valójában csak a gyakorlatban soha nem elérhet határeset, de ettl még a rá felépített modell a hozzá közel álló esetekre (pl. elég sok tizedesjeggyel sorsolt véletlenszámokkal) jól kezelheti azokat.

Kidolgozott példafeladatok

1. Egydimenziós példa

Osszuk két részre a [0,1] intervallumot egy rajta véletlenszeren választott x számmal.

Mekkora valószínséggel lesz a két rész négyzetösszege $\frac{5}{8}$-nál kisebb?

Megoldás

```
> restart;
```

```
> hatar := 5/8; evalf(%);
```

A választott x függvényében a vizsgált négyzetösszeget a következ függvény írja le:

```
> negyzetosszeg := x -> x ^ 2 + (1 - x) ^ 2;
```

Rövid kísérletezés után találunk olyan értéket, amire a négyzetösszeg a határ felett van, és olyat is, amire alatta:

```
> negyzetosszeg(0.2);
  negyzetosszeg(0.6);
```

Ábrázolhatjuk is a négyzetösszeg alakulását x függvényében, egyszerre megjelenítve a kiszemelt határt is:

```
> plot([negyzetosszeg(x), hatar], x = 0..1, y = 0..1,
  scaling=constrained);
```

Így már láthatjuk, hogy a négyzetösszeg éppen a felezésnél lesz minimális, és annál nagyobb, minél jobban eltér a kisorsolt x a felezponttól. Hogy éppen hol lép $\frac{5}{8}$ alá, az a metszéspontok megkeresésével - egy másodfokú egyenltlenség megoldásával megtudható:

```
> solve(negyzetosszeg(x)>hatar);
```

Ezek szerint akkor kisebb a négyzetösszeg $\frac{5}{8}$-nál, ha x-et az $\left[\frac{1}{4}, \frac{3}{4}\right]$ intervallumból $[0, 1]$ intervallum hosszához arányítva kapjuk a keresett valószínséget:

```
> p := (3/4 - 1/4) / (1 - 0);
```

Végezzünk is el véletlenszám-generátor segítségével egy kísérletsorozatot, melyben $[0, 1]$-beli véletlenszámokat generálunk, és megszámoljuk azon eseteket, melyekben a négyzetösszeg a határ alatt van. Először létrehozunk egy $[0, 1]$ intervallumon egyenletes eloszlású valószínségi változót:

```
> restart:
  with(Statistics):
  X := RandomVariable(Uniform(0,1));
```

Aztán ezzel végrehajtjuk a kísérletet valahányszor:

```
> n := 1000;
  veletlenszamok := Sample(X, n):
  hatar := 5 / 8;
  negyzetosszeg := x -> x ^ 2 + (1 - x) ^ 2;
```

Írunk egy ciklust, mely megszámolja a feltételeknek megfelel értékeket, majd kiszámítjuk ezek arányát az összes kísérleten belül:

```
> jo := 0:
  for i from 1 to n do
    if(negyzetosszeg(veletlenszamok[i]) < hatar) then
      jo := jo + 1:
    end if:
  end do:
```

```
> jo;
  evalf(jo / n, 3); # 3 tizedesjegy potossággal íratjuk ki
```

Látszik, hogy a kísérletben kapott érték közel esik az elméleti valószínséghez.

Külön tanulmányozhatjuk a kísérletek során kapott négyzetösszegek eloszlását is. Helyezzük el egy listában az értékeket:

```
> negyzetosszegek := [seq(negyzetosszeg(veletlenszamok[i]
  ), i=1..n)]:
```

Ábrázoljuk a kísérletsorozatban kapott értékeket:

```
> pontok := [seq([i, negyzetosszegek[i]], i=1..n)]:
```

```
> plot(pontok, style=point);
```

```
> Histogram(negyzetosszegek, bincount=16);
```

Mindkét ábrán jól érzékelhet, hogy a [0.5, 1]-beli értékek a 0.5 felli oldalon srbbek, és a másik szél felé haladva ritkulnak. Ez összhangban áll azzal, hogy fentebbi számításunk szerint a négyzetösszeg-értékek fele 0.5 és 5/8~0.625 közé (b $\frac{1}{8}$ hosszúságú intervallumba) esik, míg a maradék fele háromszor olyan hosszú ($\frac{3}{8}$) szakaszon oszlik el.

2. Kétdimenziós példa

Válasszunk két véletlen értéket a [0, 1] intervallumból: x, y. Mekkora valószínséggel lesz a két érték egymáshoz 0.5-nél közelebb?

Megoldás

Ha pl. $x = 0.2$ és $y = 0.3$, akkor teljesül a feltétel:

```
> x := 0.2; y := 0.3; y - x < 0.5;
```

$x = 0.1$ és $y = 0.7$ mellett viszont már nem:

```
> x := 0.1; y := 0.7; y - x > 0.5;
```

x és y független választása a $[0, 1]$ intervallumból modellezhet egy (x, y) pont választásával a $[0, 1] \times [0, 1]$ négyzetbl. Az x és y között felírt vizsgált feltétel koordinátageometriai értelmezéssel egy geometriai feltétellé alakítható.
Az $|y - x| < 0.5$ feltétel mint egyenltlenség megoldáshalmaza a koordinátasík egy tartománya.

```
> restart:
  with(plots):
  with(Statistics):
> inequal({y - x < 0.5, x - y < 0.5}, x = 0..1, y = 0..1,
  optionsfeasible=(color=yellow), optionsexcluded=(color=
  white), scaling=constrained);
```

Az ábrán színezett tartomány pontjainak (x, y) koordinátái teljesítik az elírt feltételt, a többiek nem. Mivel az (x, y) pontok egyenletes, véletlenszer kiválasztását az alkalmazott analógia szerint az egységnégyzetre pontok egyenletes szórásának képzelhetjük, a feltételt teljesít számpárok elállásának valószínsége a pontok színezett területre való esésének valószínségével egyezik meg - márpedig a színezett terület részesedése az egységnégyzetbl 75%.

```
> tOmega := 1*1;
  tA := tOmega - 2*(0.5*0.5/2);
  PA := tA/tOmega;
```

Végrehajthatunk egy véletlenszám-generátoros kísérletet is:

```
> n := 10000;
  X := RandomVariable(Uniform(0,1)):
> veletlen_X := Sample(X, n):
  veletlen_Y := Sample(X, n):
  veletlen_X[1]; veletlen_Y[1];
> tavolsag := (x, y) -> abs(x - y);
> jo := 0:
  for i from 1 to n do
    if(tavolsag(veletlen_X[i], veletlen_Y[i]) < 0.5) then
      jo := jo + 1:
    end if:
  end do:
> jo;
  evalf(jo / n, 3);
```

A kísérletileg kapott érték az elméletileg számított érték közelében van.

3. Másodfokú egyenlet gyökei

Válasszunk egy B és egy C számot véletlenszeren, egymástól függetlenül a $[-1, 1]$ intervallumban és tekintsük az $x^2 + B\,x + C = 0$ másodfokú egyenletet! Keressük meg annak a valószínségét, hogy az egyenlet mindkét gyöke
a) valós;
b) pozitív!

Megoldás

```
> restart;
```

a)

Az eseménytér: $\Omega = \{(B, C) \mid B \in [-1, 1], C \in [-1, 1]\} = [-1, 1]^2$, ami egy 2 oldalhosszúságú négyzet. Mivel B-t és C-t egymástól függetlenül, véletlenszeren (egyenletes eloszlással) választjuk a $[-1, 1]$ intervallumból, ezért a valószínségi mez geometriai, azaz Ω bármely részhalmazának valószínsége arányos a részhalmaz geometriai mértékével (területével).
Ω területe:

```
> TOmega := 2*2;
```

Az egyenletünk:

```
> eq := x^2+B*x+C = 0;
```

Jelölje A azt az eseményt, hogy ennek a másodfokú egyenletnek minden megoldása valós. Ez pontosan akkor következik be, ha a diszkrimináns nemnegatív:

```
> Dis := B^2-4*1*C;
> ineq_real := isolate(Dis >= 0, C); # C-re rendezzük az egyenltlenséget
```

Azaz a kedvez esetek halmaza Ω-nak a $B \to \frac{1}{4} B^2$ egyenlet parabola alatti tartománya lesz:

```
> with(plots):
  plot([rhs(ineq_real), -1], B=-1..1, C=-1..1, color=[blue, blue], thickness=3, scaling=constrained, gridlines=true, font = [axes, "HELVETICA", 16], filled=true);
```

Területét integrálással határozhatjuk meg:

```
> TA := 2*1 + int(rhs(ineq_real), B=-1..1); # x-tengely alatti + görbe és x-tengely közötti terület; az rhs parancs az egyenltlenség jobb oldalát adja vissza
```

Az A esemény valószínsége a geometriai kiszámítási mód alapján:

```
> PA := TA/TOmega; evalf(%);
```

b)

Legyen $A2$ az az esemény, hogy a fenti egyenlet minden gyöke pozitív.

```
> x1, x2 := solve(eq, x);
```

Látható, hogy ha van (valós) megoldás, akkor $\sqrt{B^2 - 4\,C} \geq 0$, így mindig x_2 a kisebbik gyök és elég annak pozitivitását vizsgálni:

```
> ineq_pos := x2 > 0;
```

Ezt az egyenltlenséget algebrai úton oldjuk meg. Elször vegyük észre, hogy $B < 0$-nak mindenképpen teljesülnie kell, különben $x_2 \leq 0$ lenne. Mivel az **??** egyenltlenség túl összetett ahhoz, hogy a Maple *solve* parancsával közvetlenül megoldhassuk, ezért lépésenkénti egyenletrendezéssel haladhatunk tovább.

```
> ineq_pos;
  % + B/2; # mindkét oldalhoz hozzáadunk B/2-t; a "%" mindig az elz egyenltlenségre hivatkozik
  %*2; # mindkét oldalt szorozzuk 2-vel
  %*(-1); # szorzunk (-1)-gyel, hogy mindkét oldal pozitív legyen (emlékezzünk, hogy B<0 volt)
  map(s->s^2, %); # mindkét oldalt négyzetre emeljük a map parancs segítségével; ez itt ekvivalens átalakítás, mivel mindkét oldal értéke pozitív
  # innentl már befejezhetnénk a solve(%, C) paranccsal is, ami a RealRange(-infinity, Open(0)) intervallumot adná vissza, de folytassuk inkább az egyenletrendezést
  % + 4*C-B^2; # mindkét oldalhoz hozzáadunk (4*C-B^2)-et
  %/4; # végül osztunk 4-gyel
```

Természetesen az a) pont feltételének is teljesülnie kell, különben nem létezne valós megoldás. Tehát a kedvez tartomány a parabola alatti terület $B < 0$ és $C > 0$ feltételeknek eleget tev része.

```
> P1 := plot(rhs(ineq_real), B=-1..1, C=-1..1, color=red, thickness=3, scaling=constrained, gridlines=true, font = [axes, "HELVETICA", 16]):
  P2 := plot(rhs(ineq_real), B=-1..0, C=-1..1, color=red, thickness=3, scaling=constrained, gridlines=true, font = [axes, "HELVETICA", 16], filled=true):
  display(P1, P2);
```

Számítsuk ki integrálással az *A2* eseményhez tartozó területet:

```
> TA2 := int(B^2/4, B=-1..0);
```

Tehát annak a valószínsége, hogy mindkét megoldás pozitív:

```
> PA2 := TA2/TOmega; evalf(%);
```

Érdekes, hogy ez csak töredéke ($\frac{1}{26}$-od része) az a) pontban kapott valószínségnek, pedig a lehetséges előjel-kombinációk alapján –naív logikával– $\frac{1}{4}$-et várnánk. Kiderült, hogy ha van megoldás, akkor jó eséllyel található nempozitív megoldás is.

4. Céltábla

Kör alakú R sugarú céltáblára lövünk (a találat valószínsége arányos a területtel). A céltáblát koncentrikus körökkel 10 részre akarjuk felosztani úgy, hogy mind a 10 részben a találat valószínsége ugyanakkora legyen. Hogyan kell a körök sugarait megválasztani?

Megoldás

```
> restart;
```

```
> R; # a feladat paramétere
```

```
> n := 10; # a részek száma
```

A céltáblát pontosan 9 darab koncentrikus körrel tudjuk 10 részre (körgyrre) felosztani. Jelölje ezek sugarát növekv sorrendben $0 < r_1 < r_2 < ... < r_9 < r_{10} = R$ (a 10. kör sugara maga R).

Ljünk rá véletlenszeren a céltáblára! Az eseménytér a céltábla pontjainak halmaza. Ha a céltáblát egy origó középpontú körnek képzeljük el a koordinátarendszerben, akkor $\Omega = \left\{ \underline{x} \in \mathbb{R}^2 \,\middle|\, |\underline{x}| \leq R \right\}$. Ha tényleg véletlenszer pontban találjuk el a céltáblát (de mindig eltaláljuk), akkor bármely részét annak területével arányos valószínséggel találjuk el, ezért geometriai valószínségi mezt feltételezhetünk.

Így ahhoz, hogy az egyes körgyrket egyenl valószínséggel találjuk el, elég az r_k $(k = 1..9)$ sugarakat úgy megválasztani, hogy a keletkez körgyrk egyenl T_{gy} területek legyenek. Mivel összesen 10 körgyr van, ezért T_{gy} a teljes terület $\frac{1}{10}$ -e:

```
> Tgy := R^2*Pi/n;
```

Az els k darab körgyr területeinek összege egyrészt $k \cdot T_{gy}$, másrészt pedig $r_k^2 \cdot \pi$. Ebbl az egyenletbl r_k meghatározható.

```
> k*Tgy = r[k]^2*Pi;
  [solve(%, r[k])];
```

Számunkra csak a pozitív megoldás az érdekes. Rögzítsük ezt egy függvényben:

```
> sugar := k -> sqrt(k/10)*R;
```

Végül listázzuk ki a körök sugarait:

```
> for k from 1 to 9 do
    r[k]:=sugar(k);
  end do;
  r[10] := R;
```

Például, ha $R = 10$, akkor a körök sugarai növekv sorrendben:

```
> R := 10:
  evalf(seq(r[k], k=1..10), 4); # az evalf parancs
  sorozatokra is mködik, az utolsó argumentum a
  megjelenítend tizedesjegyek száma
```

Ábrázoljuk a céltáblát koordinátarendszerben. Érdemes polárkoordinátákat használni.

```
> with(plots):
  P1 := polarplot({seq(r[k], k=1..9)}, theta=0..2*Pi,
  color=red, thickness=2, gridlines=false, axes=none):
  P2 := polarplot({0.1, 10}, theta=0..2*Pi, color=black,
  thickness=4, gridlines=false, axes=none): # jelöljük a
  céltábla szélét, illetve a középpontját is egy kicsi,
  nemnulla sugarú körrel
```

```
display({P1, P2}); # a két plot kirajzolása közös
koordinátarendszerben
```

5. Kerti locsoló

Egy 50 m × 30 m-es kert négy sarkában egy-egy locsoló van elhelyezve. Mekkora valószínséggel kap vizet egy véletlenszeren kiválasztott vetemény, ha a locsolók hatósugara

a) 10 m;

b) 15 m;

c) 25 m?

Megoldás

```
> restart: with(plots):
```

a)

```
> a, b := 50, 30; # a kert méretei
  r1 := 10; # a locsolók hatósugara
```

Rajzoljuk fel a veteményest és a locsolók hatókörét:

```
> with(plots, implicitplot):
  P0 := inequal({0<=x, x<=a, 0<=y, y<=b}, x=0..a, y=0..b,
  scaling=constrained, font = [axes, "HELVETICA", 16],
  optionsfeasible = [color="White"]):
  P1_a := implicitplot(x^2+y^2=r1^2, x=0..a, y=0..b,
  color=red, scaling=constrained, font = [axes,
  "HELVETICA", 16]):
  P2_a := implicitplot((x-a)^2+y^2=r1^2, x=0..a, y=0..b,
  color=red, scaling=constrained, font = [axes,
  "HELVETICA", 16]):
  P3_a := implicitplot(x^2+(y-b)^2=r1^2, x=0..a, y=0..b,
  color=red, scaling=constrained, font = [axes,
  "HELVETICA", 16]):
  P4_a := implicitplot((x-a)^2+(y-b)^2=r1^2, x=0..a, y=0..
  b, color=red, scaling=constrained, font = [axes,
  "HELVETICA", 16]):
  display(P0, P1_a, P2_a, P3_a, P4_a);
```

Válasszunk ki véletlenszeren egy veteményt a kertben. Az eseménytér a kert pontjainak halmaza: $\Omega = \{ (x, y) \in \mathbb{R}^2 \mid 0 \le x \le 50, 0 \le y \le 30 \} = [0, 50] \times [0, 30]$. Területe:

```
> TOmega := a*b;
```

Jelölje A azt az eseményt, hogy a kiválasztott növény nem lesz meglocsolva! Ez az alábbi ábrán a szürke területnek felel meg.




```
> Tk_a := r1^2*Pi/4; # egy negyedkör területe
  TA := TOmega - 4*Tk_a; # az öntözetlen rész területe
```

A geometriai kiszámítási mód alapján a keresett valószínség:

```
> PA := TA/TOmega; evalf(%);
```

Tehát körülbelül 10 növénybl 8 nem kap vizet!

b)

```
> r2 := 15;
```

Rajzoljuk fel a veteményest és a locsolók hatókörét!

```
> with(plots, implicitplot):
  P0 := inequal({0<=x, x<=a, 0<=y, y<=b}, x=0..a, y=0..b,
  scaling=constrained, font = [axes, "HELVETICA", 16],
  optionsfeasible = [color="White"]):
  P1_a := implicitplot(x^2+y^2=r2^2, x=0..a, y=0..b,
  color=red, scaling=constrained, font = [axes,
  "HELVETICA", 16]):
  P2_a := implicitplot((x-a)^2+y^2=r2^2, x=0..a, y=0..b,
  color=red, scaling=constrained, font = [axes,
  "HELVETICA", 16]):
  P3_a := implicitplot(x^2+(y-b)^2=r2^2, x=0..a, y=0..b,
  color=red, scaling=constrained, font = [axes,
  "HELVETICA", 16]):
  P4_a := implicitplot((x-a)^2+(y-b)^2=r2^2, x=0..a, y=0..
  b, color=red, scaling=constrained, font = [axes,
```

```
  "HELVETICA", 16]):
  display(P0, P1_a, P2_a, P3_a, P4_a);
```

Jelölje B azt az eseményt, hogy egy véletlenszeren kiválasztott vetemény nem lesz meglocsolva! Ez ismét a körökön kívül es rész területének felel meg.




```
> Tk_b := r2^2*Pi/4; # egy negyedkör területe
  TB := TOmega - 4*Tk_b; # a locsolatlan rész területe
```

A geometriai kiszámítási mód alapján a keresett valószínség:

```
> PB := TB/TOmega; evalf(%);
```

Ebben az esetben már majdnem a növények fele kap vizet.

c)

```
> r3 := 25;
```

Rajzoljuk fel ismét a veteményest és a locsolók hatókörét!

```
> with(plots, implicitplot):
  P0 := inequal({0<=x, x<=a, 0<=y, y<=b}, x=0..a, y=0..b,
  scaling=constrained, font = [axes, "HELVETICA", 16],
  optionsfeasible = [color="White"]):
  P1_a := implicitplot(x^2+y^2=r3^2, x=0..a, y=0..b,
  color=red, scaling=constrained, font = [axes,
  "HELVETICA", 16]):
  P2_a := implicitplot((x-a)^2+y^2=r3^2, x=0..a, y=0..b,
  color=red, scaling=constrained, font = [axes,
  "HELVETICA", 16]):
  P3_a := implicitplot(x^2+(y-b)^2=r3^2, x=0..a, y=0..b,
```

```
color=red, scaling=constrained, font = [axes,
"HELVETICA", 16]):
P4_a := implicitplot((x-a)^2+(y-b)^2=r3^2, x=0..a, y=0..
b, color=red, scaling=constrained, font = [axes,
"HELVETICA", 16]):
display(P0, P1_a, P2_a, P3_a, P4_a);
```

Jelölje C azt az eseményt, hogy egy véletlenszeren kiválasztott vetemény nem lesz meglocsolva! Ez a középen megmaradt, rombuszszer területnek felel meg. Most körültekinten kell számolnunk, mert a locsolt területek átfednek.




```
> Tk_c := r3^2*Pi/4; # egy negyedkör területe
```

Jelölje T_2 azt a területet, amely két locsoló által is meg van öntözve (sötétebb piros terület az ábrán). Ekkor a szürke (öntözetlen) rész területe:

```
> TC := TOmega - 4*Tk_c + T2; # a sötétebb piros rész
  területét kétszer vontuk le, ezért a végén még hozzá
  kell adni egyszer
```

Hátramaradt még T_2 meghatározása. Tekintsük a következ ábrákat:



D
E
T_c
α
O
y
x
0
10
20
30
40
50



D
F
E
T_s
T_h
α
O
y
x
0
10
20
30
40
50

Jelölje T_c, T_h és T_s rendre az ODE körcikk (zöld), az OEF derékszög háromszög (cián) és az EFD fél-körszelet (sárga) területét. Ekkor,

$$\boldsymbol{T_2 = 4 \cdot T_s = 4 \cdot (T_c - T_h)}.$$

Tudjuk, hogy $OE = r_3$ és $OF = \frac{b}{2}$. Jelölje α az $FOE_{\measuredangle}$ szöget! Ekkor

$\cos\alpha = \frac{OF}{OE}$, $T_c = \frac{r_3^2 \cdot \alpha}{2}$ és $T_h = \frac{OE \cdot OF \cdot \sin\alpha}{2}$.

```
> OE := r3;
  OF := b/2;
  alpha := arccos(OF/OE); # a koszinusz függvény 
  inverzével számolunk
```

```
> Tc := r3^2*alpha/2;
  Th := OF*OE*sin(alpha)/2;
  Ts := Tc-Th;
  T2 := 4*Ts;
```

```
> evalf(TC); # ellenrizzük, hogy értelmes eredményt 
  kaptunk-e a nem locsolt területre
```

A geometriai kiszámítási mód alapján a keresett valószínség:

```
> PC := TC/TOmega; evalf(%);
```

Így már csak a növények kb. 6.4 %-a marad öntözetlenül!

6. Téglatest átlója

A $[0, 1]$ intervallumban válasszunk véletlenszeren három számot, X, Y és Z-t. Mekkora a valószínsége, hogy az X, Y és Z oldalhosszúságú téglatest testátlója kisebb 1-nél?

Megoldás

```
> restart;
```

Az eseménytér: $\Omega = \{(X, Y, Z) \in \mathbb{R}^3 \mid 0 \le X, Y, Z \le 1\} = [0, 1]^3$, ami a 3-dimenziós egységkockának felel meg. Mivel mindhárom számot véletlenszeren (geometriai valószínség szerint) választottuk a $[0, 1]$ intervallumból, ezért geometriai valószínségi mezvel van dolgunk. Az eseménytér térfogata:

```
> VOmega := 1*1*1;
```

Az X, Y és Z oldalhosszúságú téglatest testátlója:

```
> e := sqrt(X^2+Y^2+Z^2);
```

Ez nem más, mint a $P(X, Y, Z)$ pont távolsága az origótól a 3-dimenziós térben, vagyis a hozzá tartozó $\underline{p}$ helyvektor hossza.

Jelölje A az $e < 1$ eseményt. Ekkor

$A = \left\{(X, Y, Z) \in \mathbb{R}^3 \,\middle|\, 0 \le X, Y, Z \le 1 \text{ } \textit{és} \sqrt{X^2 + Y^2 + Z^2} < 1\right\} = \left\{\underline{p} \in \Omega \,\middle|\, |\underline{p}| < 1\right\}$.

```
> with(plottools): with(plots):
  plot3d(sqrt(1-x^2-y^2), x=0..1, y=0..1, filled=
  [transparency=0.5], style=surface);
```

Az A halmaz az egységgömb pozitív térnyolcadba es része, ezért térfogata az egységgömb térfogatának $\frac{1}{8}$ -a:

```
> r := 1;
  Vgömb := 4*r^3*Pi/3;
  VA := Vgömb/8;
```

A keresett valószínséget a két térfogat aránya adja:

```
> PA := VA/VOmega; evalf(%);
```

Tehát nagyjából 52.4 % eséllyel lesz a kapott téglatest testátlója kisebb 1-nél.

7. Körlap ejtése papírlapra

Párhuzamos egyeneseket húzunk egy (kellen nagy méret) papírlapra. A szomszédos egyenesek egymástól váltakozva 2 cm ill. 8 cm távolságra vannak. Véletlenszeren leejtünk egy $R = 2.5$ cm sugarú körlapot a papírlapra úgy, hogy a kör középpontja a sík bármely tartományába annak területével arányos valószínséggel esik. Mekkora a valószínsége, hogy a körlap egyik egyenesbe se metsz bele?

Megoldás

```
> restart: with(plots):
```

A feladat érezheten hasonlít a Buffon-féle tproblémához, de mint látni fogjuk, annál lényegesen egyszerbb a megoldása.
Mivel a papírlap mintázata periodikus, ezért elég egy periódust vizsgálni, vagyis egy 10 cm széles tartományt, melyet mindkét szélén egyenes határol, és tartalmaz egy harmadik egyenest, mely a tartomány jobb ill. a bal szélétl befelé 2 cm ill. 8 cm távolságra van. Feltesszük, hogy a körlap középpontja ebbe a tartományba esik.

Jelölje x a körlap középpontjának a tartomány jobb szélétl vett távolságát és jelölje A azt az eseményt, hogy a leejtett körlap belemetsz valamely egyenesbe a papírlapon. Vegyük észre, hogy az A esemény bekövetkezése csak x értékétl függ, így a feladatot modellezhetjük a számegyenesen.
Az eseménytér $\Omega = [0, 10]$. A párhuzamos egyenesek az $x_0 = 0$, $x_1 = 2$ és $x_2 = 10$ pontokban (távolságokban) helyezkednek el, a kedvez esemény bekövetkezéséhez pedig az kell, hogy x-tl mindhárom pont több, mint 2.5 cm távolságra legyen. Ez alapján könnyen belátható, hogy $A = [4.5, 7.5]$.
Mivel a ledobás során x véletlenszeren (egyenletes eloszlással) sorsolódik a $[0, 10]$ intervallumból (nincs okunk azt feltételezni, hogy bármely távolságban "srsödést" tapasztalnánk a leejtett körlapok számában), ezért geometriai valószínségi mezvel van dolgunk, esemény valószínségét az esemény és az eseménytér geometriai mértékének aránya adja:

$$P(A) = \frac{m(A)}{m(\Omega)} = \frac{7.5 - 4.5}{10 - 0} = \frac{3}{10} = 0.3$$

```
> mOmega := 10;
  mA := 7.5 - 4.5;
```

```
> PA := mA/mOmega;
```

Tehát 30 % az esély arra, hogy a körlap nem találkozik egyenessel a papírlapon.

8. Téglalap területe

Egy 10 egység hosszúságú szakaszt egy rajta véletlenszeren választott ponttal kettéosztunk.
a) Ha a létrejöv darabokat egy téglalap oldalainak tekintjük, mekkora lehet a minimális és a maximális terület?
b) Mekkora valószínséggel esik a terület a lehetséges tartomány fels felébe?
c) Végezzünk véletlenszám-generátoros kísérletet a feladatra (a megfelel esetek ciklusban való összeszámlálásával), és az eredményt vessük össze az elméletileg számított értékkel!

Megoldás

```
> restart;
```

a)

```
> h := 10; # eredeti szakasz hossza
```

Jelölje x az osztópontot $(0 \le x \le 10)$. Egy x hosszú és egy $10 - x$ hosszú szakasz keletkezik.

```
> with(Statistics):
```

Az eseménytér: $\Omega = [0, 10]$. Geometriai mértéke:

```
> mOmega := h;
```

A téglalap területe:

```
> terulet := x -> x*(10-x);
```

```
> terulet(4);
```

A minimális terület 0, mert x akármilyen kicsinek választható, de $10 - x$ korlátos mennyiség. (vagy: $x = 0$-ra 0 a terület és negatív nem lehet, tehát 0 a legkisebb).

A maximális terület:

```
> T_max := maximize(terulet(x));
  T_min := 0;
```

b)

A terület lehetséges tartományának fels fele:

```
> T_fel := T_max/2;
```

Jelöljük A-val azt az eseményt, hogy a terület T_{fel} és $T_{\max}$ közé esik. $P(A)$ a kérdés. Milyen x-ekre teljesül, hogy $T_{fel} \le terulet \le T_{\max}$.

```
> plot([T_fel, terulet(x)], x = 0..10);
```

A grafikonon az az intervallum jó, ahol a parabola a vízszintes vonal felett van.

```
> L := [solve(terulet(x) = T_fel)];
```

Az A esemény geometriai mértéke az intervallum két végpontjának különbsége:

```
> mA := L[2] - L[1];
```

A keresett valószínség a geometriai kiszámítási mód alapján:

```
> pA := mA/mOmega;
  evalf(%);
```

Tehát nagyjából 70.7 $\%$ eséllyel lesz a terület a lehetséges tartomány fels felében.

c)

```
> X := RandomVariable(Uniform(0, 10));
```

```
> Probability(X*(10-X) >= T_fel);
```

```
> n := 1000; # a kísérletek száma
> minta := Sample(X, n);
> counter := 0;
> esemenyA := proc(x)
    return evalb(x*(10-x) >= T_fel);
  end proc;
> esemenyA(1);
> V := Vector(n);
> for i from 1 to n do
    x := minta[i]:
    if esemenyA(x) then
      counter := counter + 1:
    end if:
    V[i] := counter/i; # relatív gyakoriság
  end do:
> V[100];
> unassign('x');
  P0 := LineChart(V, gridlines = true, axes = boxed,
  symbolsize = 3):
  P1 := plot(pA, x=1..n, y=0..1):
  plots[display](P0, P1);

> evalf(V[n]);
```

Jól látható, hogy a relatív gyakoriságok az elméleti valószínséghez konvergálnak.

9. Vitorlarúd

Hajótöröttek egy lakatlan szigeten azt tervezik, hogy a viharban összeroncsolódott hajójuk árboc rúdjának felhasználásával új, kisebb hajót építenek. A vihar a 40 méteres árbocrudat három véletlenszer darabra törte szét. Ha van olyan darab, amelyik 20 méternél hosszabb, akkor a kisebb hajót meg tudják építeni. Mekkora valószínséggel találnak olyan hosszú darabot, hogy a hajó megépíthet legyen, amikor kiúsznak a roncsokhoz?

Megoldás

```
> restart;
```

Legyen a három darab hossza rendre x, y, z. A három darab összesen 40 méter hosszú:

$$x + y + z = 40.$$

Ahonnan $z = 40 - x - y$ és így elegend az (x, y) független változó-párral leírni az összes esetet. Mivel $z \geq 0$, ezért $x + y \leq 40$. Tehát az eseménytér:

$$\Omega = \{ (x, y) \mid 0 \leq x \leq 40, 0 \leq y \leq 40, x + y \leq 40 \}$$

```
> with(plots):
> E := inequal(x + y <= 40, x = 0..40, y = 0..40, color =
  grey): E;
```

Az eseménytér egy derékszög háromszög, melynek területmértéke

$$m(\Omega) = \frac{40 \cdot 40}{2} = 800.$$

```
> mOmega := 40*40/2;
```

A $z = 40 - x - y \geq 20$ az $A_z = \{x + y \leq 20\}$ halmazon teljesül, melynek területe

$$m(A_z) = \frac{20 \cdot 20}{2} = 200.$$

```
> F1 := inequal({x + y <= 40, x + y < 20}, x = 0..40, y =
  0..40, color = red, transparency = 0.5):
  display(F1, E);
```

Az $A_x = \{x \geq 20\}$ esemény teljesül az alábbi háromszögben, melynek területes

$$m(A_y) = \frac{20 \cdot 20}{2} = 200.$$

```
> F2 := inequal({x >= 20, x + y <= 40}, x = 0..40, y = 0.
  .40, color = red, transparency = 0.5):
  display(F2, E);
```

Az $A_y = \{y \geq 20\}$ esemény teljesül az alábbi halmazon, melynek területe

$$m(A_y) = \frac{20 \cdot 20}{2} = 200.$$

```
> F3 := inequal({y >= 20, x + y <= 40}, x = 0..40, y = 0.
  .40, color = red, transparency = 0.5):
  display(F3, E);
```

Akkor tudnak építeni kisebb hajót, ha teljesül az

$$A = A_z + A_x + A_y$$

esemény.

```
> display(F1, F2, F3, E); # az A esemény
```

Ezért $m(A) = 200 + 200 + 200 = 600$.

```
> mAz := 20*20/2;
  mAx := 20*20/2;
  mAy := 20*20/2;
  mA := mAx + mAy + mAz;
> PA := mA/mOmega; evalf(%);
```

Tehát 75 %-os valószínséggel találnak elég hosszú árboc rudat.
Figyeljük meg, hogy az itt kedvez esemény pont a 7. feladatban szerepl ellentettje: akkor lenne készíthet háromszög a három darabból, ha mindegyik rövidebb lenne 20 méternél!

Gyakorló feladatok

Gy1.

Válasszunk két véletlen értéket a $[0, 1]$ intervallumból: x, y.
a) Mekkora valószínséggel lesz nagyobb a második szám az els köbénél?
b) Végezzünk véletlenszám-generátoros kísérletet a feladatra (a megfelel esetek ciklusban

való összeszámlálásával), és az eredményt vessük össze az elméletileg számított értékkel!

Gy2.

Téglalapot szerkesztünk a $[0, 10]$ intervallumból egymástól függetlenül véletlenszeren választott két értékkel mint oldalhosszakkal.
a) Mekkora valószínséggel lesz a kerülete 25 egységnél nagyobb?
b) Mekkora valószínséggel lesz a területe 25 egységnél kisebb?
c) Mekkora valószínséggel lesz az átlója 10 egységnél hosszabb?
d) Mekkora valószínséggel lesz az átlója 12 egységnél rövidebb?

Gy3.

Válasszunk két véletlen értéket a $[-2, 2]$ intervallumból: x, y.
a) Mennyi a valószínsége, hogy az els szám négyzete kisebb a második számnál?
b) Mennyi a valószínsége, hogy a második szám négyzete kisebb az els számnál?
c) Mennyi a valószínsége, hogy valamelyik szám négyzete kisebb a másik számnál?